

Computer-assisted Reconstruction and Assessment of E. J. Lowe’s Modal Ontological Argument

David Fuenmayor¹ and Christoph Benzmüller^{2,1}

¹Freie Universität Berlin, Germany

²University of Luxembourg, Luxembourg

December 14, 2021

Abstract

Computers may help us to understand –not just verify– philosophical arguments. By utilizing modern proof assistants in an iterative interpretive process, we can reconstruct and assess an argument by fully formal means. Through the mechanization of a variant of St. Anselm’s ontological argument by E. J. Lowe, which is a paradigmatic example of a natural-language argument with strong ties to metaphysics and religion, we offer an ideal showcase for our computer-assisted interpretive method.

Contents

| | | |
|----------|--|----------|
| 1 | Embedding of Quantified Modal Logic | 2 |
| 1.1 | Type Declarations | 2 |
| 1.2 | Logical Constants as Truth-Sets | 2 |
| 1.3 | Quantification | 2 |
| 1.4 | Equality | 3 |
| 1.5 | Validity | 3 |
| 1.6 | Verifying the Embedding | 3 |
| 1.7 | Axiomatization of Further Logics | 4 |
| 2 | E. J. Lowe’s Modal Ontological Argument | 4 |
| 2.1 | Introduction | 4 |
| 2.2 | Initial Formalization | 6 |
| 2.3 | Validating the Argument I | 11 |
| 2.4 | Validating the Argument II | 12 |
| 2.5 | Simplifying the Argument | 13 |
| 2.6 | Arriving at a Non-Modal Argument | 14 |
| 2.7 | Modified Modal Argument I | 16 |
| 2.8 | Modified Modal Argument II | 17 |

1 Embedding of Quantified Modal Logic

As is well known, the Isabelle proof assistant [10] does not natively support modal logics, so we have used a technique known as *shallow semantic embedding*, which allows us to take advantage of the expressive power of higher-order logic in order to embed the semantics of an object language. We draw on previous work on the embedding of multimodal logics in HOL [2], which has successfully been applied to the analysis and verification of ontological arguments (e.g. [4, 3, 7]).

1.1 Type Declarations

typedecl e — Type for entities
typedecl w — Type for worlds
type-synonym $wo = w \Rightarrow bool$ — Type for world-dependent formulas

1.2 Logical Constants as Truth-Sets

Using the technique of *shallow semantic embedding* each operator gets defined as a function on world-dependent formulas or *truth sets*.

abbreviation $mand::wo \Rightarrow wo \Rightarrow wo$ (**infix** \wedge)
 where $\varphi \wedge \psi \equiv \lambda w. (\varphi w) \wedge (\psi w)$
abbreviation $mor::wo \Rightarrow wo \Rightarrow wo$ (**infix** \vee)
 where $\varphi \vee \psi \equiv \lambda w. (\varphi w) \vee (\psi w)$
abbreviation $mimp::wo \Rightarrow wo \Rightarrow wo$ (**infix** \rightarrow)
 where $\varphi \rightarrow \psi \equiv \lambda w. (\varphi w) \rightarrow (\psi w)$
abbreviation $mequ::wo \Rightarrow wo \Rightarrow wo$ (**infix** \leftrightarrow)
 where $\varphi \leftrightarrow \psi \equiv \lambda w. (\varphi w) \leftrightarrow (\psi w)$
abbreviation $mnot::wo \Rightarrow wo$ (**infix** \neg)
 where $\neg \varphi \equiv \lambda w. \neg(\varphi w)$

We embed a modal logic K by defining the box and diamond operators using restricted quantification over the set of ‘accessible’ worlds (using an *accessibility* relation R as a guard).

consts $R::w \Rightarrow w \Rightarrow bool$ (**infix** r) — Accessibility relation
abbreviation $mbox::wo \Rightarrow wo$ (**infix** \Box)
 where $\Box \varphi \equiv \lambda w. \forall v. (w r v) \rightarrow (\varphi v)$
abbreviation $mdia::wo \Rightarrow wo$ (**infix** \Diamond)
 where $\Diamond \varphi \equiv \lambda w. \exists v. (w r v) \wedge (\varphi v)$

1.3 Quantification

Quantifiers are defined analogously.

abbreviation $mforall::('t \Rightarrow wo) \Rightarrow wo$ (**infix** \forall)
 where $\forall \Phi \equiv \lambda w. \forall x. (\Phi x w)$
abbreviation $mexists::('t \Rightarrow wo) \Rightarrow wo$ (**infix** \exists)

where $\exists \Phi \equiv \lambda w. \exists x. (\Phi x w)$
abbreviation $mforallB :: ('t \Rightarrow wo) \Rightarrow wo$ (**binder** \forall)
where $\forall x. (\varphi x) \equiv \forall \varphi$
abbreviation $mexistsB :: ('t \Rightarrow wo) \Rightarrow wo$ (**binder** \exists)
where $\exists x. (\varphi x) \equiv \exists \varphi$

1.4 Equality

Two different definitions of equality are given. The first one is an extension of standard equality for use in world-dependent formulas. The second is the well-known Leibniz equality.

abbreviation $meq :: 't \Rightarrow 't \Rightarrow wo$ (**infix** \approx)
where $x \approx y \equiv \lambda w. x = y$
abbreviation $meqL :: e \Rightarrow e \Rightarrow wo$ (**infix** \approx^L)
where $x \approx^L y \equiv \lambda w. \forall \varphi. (\varphi x w) \longrightarrow (\varphi y w)$

1.5 Validity

Validity is defined as truth in *all* worlds and represented by wrapping the formula in special brackets ($\lfloor - \rfloor$).

abbreviation $valid :: wo \Rightarrow bool$ ($\lfloor - \rfloor$) **where** $\lfloor \psi \rfloor \equiv \forall w. (\psi w)$

1.6 Verifying the Embedding

The above definitions introduce modal logic K with quantification, as evidenced by the following tests.

lemma K : $\lfloor (\Box(\varphi \rightarrow \psi)) \rightarrow (\Box\varphi \rightarrow \Box\psi) \rfloor$ **by** *simp* — Verifying K principle

lemma NEC : $\lfloor \varphi \rfloor \Longrightarrow \lfloor \Box\varphi \rfloor$ **by** *simp* — Verifying *necessitation* rule

Local consequence implies global consequence (not the other way round).

lemma *localImpGlobalCons*: $\lfloor \varphi \rightarrow \xi \rfloor \Longrightarrow \lfloor \varphi \rfloor \longrightarrow \lfloor \xi \rfloor$ **by** *simp*

lemma $\lfloor \varphi \rfloor \longrightarrow \lfloor \xi \rfloor \Longrightarrow \lfloor \varphi \rightarrow \xi \rfloor$ **nitpick oops** — Countersatisfiable

(Converse-)Barcan formulas are validated in this embedding.

lemma $\lfloor (\forall x. \Box(\varphi x)) \rightarrow \Box(\forall x. (\varphi x)) \rfloor$ **by** *simp*

lemma $\lfloor \Box(\forall x. (\varphi x)) \rightarrow (\forall x. \Box(\varphi x)) \rfloor$ **by** *simp*

β -redex is valid.

lemma $\lfloor (\lambda \alpha. \varphi \alpha) (\tau :: w \Rightarrow e) \leftrightarrow (\varphi \tau) \rfloor$ **by** *simp*

lemma $\lfloor (\lambda \alpha. \varphi \alpha) (\tau :: e) \leftrightarrow (\varphi \tau) \rfloor$ **by** *simp*

lemma $\lfloor (\lambda \alpha. \Box \varphi \alpha) (\tau :: w \Rightarrow e) \leftrightarrow (\Box \varphi \tau) \rfloor$ **by** *simp*

lemma $\lfloor (\lambda \alpha. \Box \varphi \alpha) (\tau :: e) \leftrightarrow (\Box \varphi \tau) \rfloor$ **by** *simp*

Modal collapse is countersatisfiable, as shown by Nitpick [6].

lemma $\lfloor \varphi \rightarrow \Box \varphi \rfloor$ **nitpick oops**

1.7 Axiomatization of Further Logics

The best-known normal logics ($K4$, $K5$, KB , $K45$, $KB5$, D , $D4$, $D5$, ...) can be obtained by combinations of the following axioms.

abbreviation T where $T \equiv \forall \varphi. \Box \varphi \rightarrow \varphi$
abbreviation B where $B \equiv \forall \varphi. \varphi \rightarrow \Box \Diamond \varphi$
abbreviation D where $D \equiv \forall \varphi. \Box \varphi \rightarrow \Diamond \varphi$
abbreviation IV where $IV \equiv \forall \varphi. \Box \varphi \rightarrow \Box \Box \varphi$
abbreviation V where $V \equiv \forall \varphi. \Diamond \varphi \rightarrow \Box \Diamond \varphi$

Instead of postulating combinations of the above axioms we make use of the well-known *Sahlqvist correspondence*, which links axioms to constraints on a model's accessibility relation (cf. [2] for further details). We show that reflexivity, symmetry, seriality, transitivity and euclideaness imply axioms T , B , D , IV , V respectively.¹

lemma reflexive $R \implies [T]$ by blast

lemma symmetric $R \implies [B]$ by blast

lemma serial $R \implies [D]$ by blast

lemma transitive $R \implies [IV]$ by blast

lemma euclidean $R \implies [V]$ by blast

lemma preorder $R \implies [T] \wedge [IV]$ by blast — S4: reflexive + transitive

lemma equivalence $R \implies [T] \wedge [V]$ by blast — S5: preorder + symmetric

2 E. J. Lowe's Modal Ontological Argument

2.1 Introduction

E. J. Lowe presented his argument in an article named "A Modal Version of the Ontological Argument", which has been published as a chapter in [9]. The structure of this argument is very representative of philosophical arguments. It features eight premises from which new inferences are drawn until arriving at a final conclusion: the necessary existence of God (which in this case amounts to the existence of some "necessary concrete being").

(P1) God is, by definition, a necessary concrete being.

(P2) Some necessary abstract beings exist.

(P3) All abstract beings are dependent beings.

(P4) All dependent beings depend for their existence on independent beings.

(P5) No contingent being can explain the existence of a necessary being.

¹Implication can also be proven in the reverse direction (which is not needed for our purposes). Using these definitions, we can derive axioms for the most common modal logics (see also [1]). Thereby we are free to use either the semantic constraints or the related *Sahlqvist* axioms. Here we provide both versions. In what follows we use the semantic constraints for improved performance.

(P6) The existence of any dependent being needs to be explained.

(P7) Dependent beings of any kind cannot explain their own existence.

(P8) The existence of dependent beings can only be explained by beings on which they depend for their existence.

We will consider in our treatment only a representative subset of the conclusions, as presented in Lowe's article.

(C1) All abstract beings depend for their existence on concrete beings. (Follows from P3 and P4 together with D3 and D4.)

(C5) In every possible world there exist concrete beings. (Follows from C1 and P2.)

(C7) The existence of necessary abstract beings needs to be explained. (Follows from P2, P3 and P6.)

(C8) The existence of necessary abstract beings can only be explained by concrete beings. (Follows from C1, P3, P7 and P8.)

(C9) The existence of necessary abstract beings is explained by one or more necessary concrete beings. (Follows from C7, C8 and P5.)

(C10) A necessary concrete being exists. (Follows from C9.)

Lowe also introduces some informal definitions which should help the reader understand the meaning of the concepts involved in his argument (necessity, concreteness, ontological dependence, metaphysical explanation, etc.). In the following discussion, we will see that most of these definitions do not bear the significance Lowe claims.

(D1) x is a necessary being := x exists in every possible world.

(D2) x is a contingent being := x exists in some but not every possible world.

(D3) x is a concrete being := x exists in space and time, or at least in time.

(D4) x is an abstract being := x does not exist in space or time.

(D5) x depends for its existence on y := necessarily, x exists only if y exists.

(D6) (For any predicates F and G) F depend for their existence on G := necessarily, F s exist only if G s exist.

We will work *iteratively* on Lowe's argument by temporarily fixing truth-values and inferential relationships among its sentences, and then, after choosing a logic for formalization, working back and forth on the formalization of its axioms and theorems by making gradual adjustments while getting automatic real-time feedback about the suitability of our changes,

vis-a-vis the argument's validity. In this fashion, by engaging in an *iterative* process of trial and error, we work our way towards a proper understanding of the concepts involved in the argument, far beyond of what a mere natural-language based discussion would allow.

2.2 Initial Formalization

We start our first iterations with a formalized version of Lowe's argument in modal logic using the semantic embedding presented in the previous section. We first turn to the formalization of premise P1: "God is, by definition, a necessary concrete being". In order to understand the concept of *necessariness* (i.e. being a "necessary being") employed in this argument, we have a look at the definitions D1 and D2 provided by Lowe. They relate the concepts of necessariness and contingency (i.e. being a "contingent being") with existence:²

(D1) *x is a necessary being := x exists in every possible world.*

(D2) *x is a contingent being := x exists in some but not every possible world.*

The two definitions above, aimed at explicating the concepts of necessariness and contingency by reducing them to existence and quantification over possible worlds, have a direct impact on the choice of a logic for formalization. They not only call for some kind of modal logic with possible-world semantics but also lead us to consider the complex issue of existence, since we need to restrict the domain of quantification at every world.

For this argument not to become trivialized, we guarded our quantifiers so they range only over those entities existing (i.e. being actualized) at a given world. This approach is known as *actualist quantification* and is implemented in our semantic embedding by defining a world-dependent meta-logical 'existence' predicate (called "actualizedAt" below), which is the one used as a guard in the definition of the quantifiers. Note that the type *e* characterizes the domain of all beings (existing and non-existing), and the type *wo* (which is an abbreviation for $w \Rightarrow bool$) characterizes sets of worlds. The term "isActualized" thus relates beings to worlds.

consts *isActualized::e \Rightarrow wo* (**infix** *actualizedAt*)

abbreviation *forallAct::(e \Rightarrow wo) \Rightarrow wo* (\forall^A)
where $\forall^A \Phi \equiv \lambda w. \forall x. (x \text{ actualizedAt } w) \longrightarrow (\Phi \ x \ w)$

abbreviation *existsAct::(e \Rightarrow wo) \Rightarrow wo* (\exists^A)
where $\exists^A \Phi \equiv \lambda w. \exists x. (x \text{ actualizedAt } w) \wedge (\Phi \ x \ w)$

²Here, the concepts of necessariness and contingency are meant as properties of beings, in contrast to the concepts of necessity and possibility which are modals. We will see later how both pairs of concepts can be related in order to validate this argument.

We also define the corresponding binder syntax below.

abbreviation $mforallActB::(e\Rightarrow wo)\Rightarrow wo$ (**binder** $\forall^A[8]g$)

where $\forall^A x. (\varphi x) \equiv \forall^A \varphi$

abbreviation $mexistsActB::(e\Rightarrow wo)\Rightarrow wo$ (**binder** $\exists^A[8]g$)

where $\exists^A x. (\varphi x) \equiv \exists^A \varphi$

We use Isabelle's Nitpick tool [6] to verify that actualist quantification validates neither the Barcan formula nor its converse.

lemma $[(\forall^A x. \Box(\varphi x)) \rightarrow \Box(\forall^A x. \varphi x)]$

nitpick oops — Countermodel found: formula not valid

lemma $[\Box(\forall^A x. \varphi x) \rightarrow (\forall^A x. \Box(\varphi x))]$

nitpick oops — Countermodel found: formula not valid

With actualist quantification in place we can: (i) formalize the concept of existence in the usual form (by using a restricted particular quantifier), (ii) formalize necessariness as existing necessarily, and (iii) formalize contingency as existing possibly but not necessarily.

definition $Existence::e\Rightarrow wo$ ($E!$) **where** $E! x \equiv \exists^A y. y \approx x$

definition $Necessary::e\Rightarrow wo$ **where** $Necessary x \equiv \Box E! x$

definition $Contingent::e\Rightarrow wo$ **where** $Contingent x \equiv \Diamond E! x \wedge \neg Necessary x$

Note that we have just chosen our logic for formalization: a free quantified modal logic K with positive semantics. The logic is *free* because the domain of quantification (for actualist quantifiers) is a proper subset of our universe of discourse, so we can refer to non-actual objects. The semantics is *positive* because we have placed no restriction regarding predication on non-actual objects, so they are also allowed to exemplify properties and relations. We are also in a position to embed stronger normal modal logics (KB , $KB5$, $S4$, $S5$, ...) by restricting the accessibility relation R with additional axioms.

Having chosen our logic, we can now turn to the formalization of the concepts of abstractness and concreteness. As seen previously, Lowe has already provided us with an explication of these concepts:

(D3) x is a concrete being := x exists in space and time, or at least in time.

(D4) x is an abstract being := x does not exist in space or time.

Lowe himself acknowledges that the explication of these concepts in terms of existence "in space and time" is superfluous, since we are only interested in them being complementary.³ Thus we start by formalizing concreteness

³We quote from Lowe's original article: "Observe that, according to these definitions, a being cannot be both concrete and abstract: being concrete and being abstract are mutually exclusive properties of beings. Also, all beings are either concrete or abstract ... the abstract/concrete distinction is exhaustive. Consequently, a being is concrete if and only if it is not abstract."

as a *primitive* world-dependent predicate and then derive abstractness from it, namely as its negation.

consts *Concrete::e⇒wo*

abbreviation *Abstract::e⇒wo* **where** *Abstract x* $\equiv \neg(\text{Concrete } x)$

We can now formalize the definition of Godlikeness (P1) as follows:

abbreviation *Godlike::e⇒wo* **where** *Godlike x* $\equiv \text{Necessary } x \wedge \text{Concrete } x$

We also formalize premise P2 ("Some necessary abstract beings exist") as shown below:

axiomatization where

P2: $[\exists^A x. \text{Necessary } x \wedge \text{Abstract } x]$

Let's now turn to premises P3 ("All abstract beings are dependent beings") and P4 ("All dependent beings depend for their existence on independent beings"). We have here three new concepts to be explicated: two predicates "dependent" and "independent" and a relation "depends (for its existence) on", which has been called *ontological dependence* by Lowe. Following our linguistic intuitions concerning their interrelation, we start by proposing the following formalization:

consts *dependence::e⇒e⇒wo* (**infix** *dependsOn*)

definition *Dependent::e⇒wo* **where** *Dependent x* $\equiv \exists^A y. x \text{ dependsOn } y$

abbreviation *Independent::e⇒wo* **where** *Independent x* $\equiv \neg(\text{Dependent } x)$

We have formalized ontological dependence as a *primitive* world-dependent relation and refrained from any explication (as suggested by Lowe).⁴

We have called an entity *dependent* if and only if there *actually exists* an object *y* such that *x depends for its existence* on it; accordingly, we have called an entity *independent* if and only if it is not dependent.

As a consequence, premises P3 ("All abstract beings are dependent beings") and P4 ("All dependent beings depend for their existence on independent beings") become formalized as follows.

axiomatization where

P3: $[\forall^A x. \text{Abstract } x \rightarrow \text{Dependent } x]$ **and**

P4: $[\forall^A x. \text{Dependent } x \rightarrow (\exists^A y. \text{Independent } y \wedge x \text{ dependsOn } y)]$

⁴An explication of this concept has been suggested by Lowe in definition D5 ("x depends for its existence on y := necessarily, x exists only if y exists"). Concerning this alleged definition, he has written in a footnote to the same article: "Note, however, that the two definitions (D5) and (D6) presented below are not in fact formally called upon in the version of the ontological argument that I am now developing, so that in the remainder of this chapter the notion of existential dependence may, for all intents and purposes, be taken as primitive. There is an advantage in this, inasmuch as finding a perfectly apt definition of existential dependence is no easy task, as I explain in 'Ontological Dependence.'" Lowe refers hereby to his article on ontological dependence in the Stanford Encyclopedia of Philosophy [8] for further discussion.

Concerning premises P5 ("No contingent being can explain the existence of a necessary being") and P6 ("The existence of any dependent being needs to be explained"), a suitable formalization for expressions of the form: "the entity X explains the existence of Y" and "the existence of X is explained" needs to be found. These expressions rely on a single binary relation, which will initially be taken as *primitive*. This relation has been called *metaphysical explanation* by Lowe.

consts *explanation::e⇒e⇒wo* (**infix** *explains*)

definition *Explained::e⇒wo* **where** *Explained x* $\equiv \exists^A y. y \text{ explains } x$

axiomatization where

P5: $\lceil \neg(\exists^A x. \exists^A y. \text{Contingent } y \wedge \text{Necessary } x \wedge y \text{ explains } x) \rceil$

Premise P6, together with the last two premises: P7 ("Dependent beings of any kind cannot explain their own existence") and P8 ("The existence of dependent beings can only be explained by beings on which they depend for their existence"), were introduced by Lowe in order to relate the concept of *metaphysical explanation* to *ontological dependence*.⁵

axiomatization where

P6: $\lceil \forall x. \text{Dependent } x \rightarrow \text{Explained } x \rceil$ **and**

P7: $\lceil \forall x. \text{Dependent } x \rightarrow \neg(x \text{ explains } x) \rceil$ **and**

P8: $\lceil \forall x y. y \text{ explains } x \rightarrow x \text{ dependsOn } y \rceil$

Although the last three premises seem to couple very tightly the concepts of (metaphysical) explanation and (ontological) dependence, both concepts are not meant by Lowe to be equivalent.⁶ We have used Nitpick in order to test this claim. Since a countermodel has been found, we have proven that the (inverse) equivalence of metaphysical explanation and ontological dependence is not implied by the axioms.

lemma $\lceil \forall x y. x \text{ explains } y \leftrightarrow y \text{ dependsOn } x \rceil$ **nitpick**_[user-axioms] **oops**

For any being, however, having its existence "explained" is equivalent to its existence being "dependent" (on some other being). This follows already from premises P6 and P8, as shown above by Isabelle's prover.

lemma $\lceil \forall x. \text{Explained } x \leftrightarrow \text{Dependent } x \rceil$

using *P6 P8 Dependent-def Explained-def* **by** *auto*

The Nitpick model finder is also useful to check axioms' consistency at any stage during the formalization of an argument. We instruct Nitpick to gen-

⁵Note that we use non-guarded quantifiers for the formalization of the last three premises in order to test argument's validity under the strongest assumptions. As before, we turn a blind eye to modal expressions like "can", "needs to", etc.

⁶Lowe says: "Existence-explanation is not simply the inverse of existential dependence. If x depends for its existence on y, this only means that x cannot exist without y existing. This is not at all the same as saying that x exists because y exists, or that x exists in virtue of the fact that y exists."

erate a model satisfying some tautological sentence (here we use a trivial ‘True’ proposition) while taking into account all previously defined axioms.

lemma *True* **nitpick**[*satisfy, user-axioms*] **oops**

In this case, Nitpick was able to find a model satisfying the given tautology; this means that all axioms defined so far are consistent. The model found has a cardinality of two for the set of individual objects and a single world.

We can also use model finders to perform ‘sanity checks’. We can instruct Nitpick to find a countermodel for some specifically tailored formula which we want to make sure is not valid. We check below, for instance, that our axioms are not too strong as to imply *metaphysical necessitism* (i.e. all beings necessarily exist) or *modal collapse*. Since both would trivially validate the argument.

lemma [$\forall x. E! x$]

nitpick[*user-axioms*] **oops** — Countermodel found: necessitism is not valid

lemma [$\varphi \rightarrow \Box\varphi$]

nitpick[*user-axioms*] **oops** — Countermodel found: modal collapse is not valid

By using Isabelle’s *Sledgehammer* tool [5], we can verify the validity of the selected conclusions C1, C5 and C7, and even find the premises they rely upon.

(C1) *All abstract beings depend for their existence on concrete beings.*

theorem *C1*: [$\forall^A x. \text{Abstract } x \rightarrow (\exists y. \text{Concrete } y \wedge x \text{ dependsOn } y)$]
using *P3 P4* **by** *blast*

(C5) *In every possible world there exist concrete beings.*

theorem *C5*: [$\exists^A x. \text{Concrete } x$]
using *P2 P3 P4* **by** *blast*

(C7) *The existence of necessary abstract beings needs to be explained.*

theorem *C7*: [$\forall^A x. (\text{Necessary } x \wedge \text{Abstract } x) \rightarrow \text{Explained } x$]
using *P3 P6* **by** *simp*

The last three conclusions are shown by Nitpick to be non-valid even in an *S5* logic. *S5* can be easily introduced by postulating that the accessibility relation *R* is an equivalence relation. This exploits the well-known *Sahlqvist correspondence* which links modal axioms to constraints on a model’s accessibility relation.

axiomatization where

S5: equivalence R — $\Box\varphi \rightarrow \varphi$, $\varphi \rightarrow \Box\Diamond\varphi$ and $\Box\varphi \rightarrow \Box\Box\varphi$

(C8) *The existence of necessary abstract beings can only be explained by concrete beings.*

lemma C8: $[\forall^A x.(Necessary\ x \wedge Abstract\ x) \rightarrow (\forall^A y. y\ explains\ x \rightarrow Concrete\ y)]$
nitpick_[user-axioms] **oops** — Countermodel found

(C9) *The existence of necessary abstract beings is explained by one or more necessary concrete (Godlike) beings.*

lemma C9: $[\forall^A x.(Necessary\ x \wedge Abstract\ x) \rightarrow (\exists^A y. y\ explains\ x \wedge Godlike\ y)]$
nitpick_[user-axioms] **oops** — Countermodel found

(C10) *A necessary concrete (Godlike) being exists.*

theorem C10: $[\exists^A x. Godlike\ x]$
nitpick_[user-axioms] **oops** — Countermodel found

By employing the Isabelle proof assistant we prove non-valid a first formalization attempt of Lowe’s modal ontological argument. This is, however, just the first of many iterations in our interpretive endeavor. Based on the information recollected so far, we can proceed to make the adjustments necessary to validate the argument. We will see how these changes have an impact on our understanding of all concepts (necessariness, concreteness, dependence, explanation, etc.).

2.3 Validating the Argument I

By examining the countermodel found by Nitpick for C10 we can see that some necessary beings that are abstract in the actual world may indeed be concrete in other accessible worlds. Lowe had previously presented numbers as an example of such necessary abstract beings. It can be argued that numbers, while existing necessarily, can never be concrete in any possible world, so we add the restriction of abstractness being an essential property, i.e. a locally rigid predicate.

axiomatization where

abstractness-essential: $[\forall x. Abstract\ x \rightarrow \Box Abstract\ x]$

theorem C10: $[\exists^A x. Godlike\ x]$

nitpick_[user-axioms] **oops** — Countermodel found

As Nitpick shows us, the former restriction is not enough to prove C10. We try postulating further restrictions on the accessibility relation R which, taken together, would amount to it being an equivalence relation. Following the *Sahlqvist correspondence*, this would make for a modal logic $S5$, and our abstractness property would consequently become a (globally) rigid predicate.

axiomatization where

T-axiom: reflexive R **and** — $\Box\varphi \rightarrow \varphi$

B-axiom: symmetric R **and** — $\varphi \rightarrow \Box\Diamond\varphi$

IV-axiom: transitive R — $\Box\varphi \rightarrow \Box\Box\varphi$

theorem *C10*: $[\exists^A x. \text{Godlike } x]$
nitpick_[user-axioms] **oops** — Countermodel found

By examining the new countermodel found by Nitpick we notice that at some worlds there are non-existent concrete beings. We want to disallow this possibility, so we make concreteness an existence-entailing property.

axiomatization where *concrete-exist*: $[\forall x. \text{Concrete } x \rightarrow E! x]$

We carry out the usual ‘sanity checks’ to make sure the argument has not become trivialized.⁷

lemma *True*

nitpick_[satisfy, user-axioms] **oops** — Model found: axioms are consistent

lemma $[\forall x. E! x]$

nitpick_[user-axioms] **oops** — Countermodel found: necessitism is not valid

lemma $[\varphi \rightarrow \Box\varphi]$

nitpick_[user-axioms] **oops** — Countermodel found: modal collapse is not valid

Since this time Nitpick was not able to find a countermodel for C10, we have enough confidence in the validity of the formula to ask Sledgehammer to search for a proof.

theorem *C10*: $[\exists^A x. \text{Godlike } x]$ **using** *Existence-def Necessary-def abstractness-essential concrete-exist P2 C1 B-axiom* **by** *meson*

Sledgehammer is able to find a proof relying on all premises but the two modal axioms *T* and *IV*. By the end of this iteration we see that Lowe’s modal ontological argument depends for its validity on three non-stated (i.e. implicit) premises: the essentiality of abstractness, the existence-entailing nature of concreteness and the modal axiom *B* ($\varphi \rightarrow \Box\Diamond\varphi$). Moreover, we have also shed some light on the meaning of the concepts of abstractness and concreteness.

2.4 Validating the Argument II

We present a slightly simplified version of the original argument (without the implicit premises stated in the previous version). In this variant premises P1 to P5 remain unchanged and none of the last three premises proposed by Lowe (P6 to P8) show up anymore. Those last premises have been introduced in order to interrelate the concepts of explanation and dependence in such a way that they play somewhat opposite roles. Now we want to go all the way and simply assume that they are inverse relations, for we want to understand how the interrelation of these two concepts affects the validity of the argument.

⁷These checks are being carried out after postulating axioms for every iteration, so we won’t mention them anymore.

axiomatization where

dep-expl-inverse: $[\forall x y. y \text{ explains } x \leftrightarrow x \text{ dependsOn } y]$

We proceed to prove the relevant partial conclusions.

theorem C1: $[\forall^A x. \text{Abstract } x \rightarrow (\exists y. \text{Concrete } y \wedge x \text{ dependsOn } y)]$
using *P3 P4* **by** *blast*

theorem C5: $[\exists^A x. \text{Concrete } x]$
using *P2 P3 P4* **by** *blast*

theorem C7: $[\forall^A x. (\text{Necessary } x \wedge \text{Abstract } x) \rightarrow \text{Explained } x]$
using *Explained-def P3 P4 dep-expl-inverse* **by** *meson*

But the final conclusion C10 is still countersatisfiable, as shown by Nitpick:

theorem C10: $[\exists^A x. \text{Godlike } x]$
nitpick[*user-axioms*] **oops** — Countermodel found

Next, we try assuming a stronger modal logic. We do this by postulating further axioms using the *Sahlqvist correspondence* and asking Sledgehammer to find a proof. Sledgehammer is in fact able to find a proof for C10 which only relies on the modal axiom *T* ($\Box\varphi \rightarrow \varphi$).

axiomatization where

T-axiom: reflexive *R* **and** $\Box\varphi \rightarrow \varphi$

B-axiom: symmetric *R* **and** $\varphi \rightarrow \Box\Diamond\varphi$

IV-axiom: transitive *R* $\Box\varphi \rightarrow \Box\Box\varphi$

theorem C10: $[\exists^A x. \text{Godlike } x]$ **using** *Contingent-def Existence-def P2 P3 P4 P5 dep-expl-inverse T-axiom* **by** *meson*

In this series of iterations we have verified a modified version of the original argument by Lowe. Our understanding of the concepts of *ontological dependence* and *metaphysical explanation* have changed after the introduction of an additional axiom constraining both: they are now inverse relations. Still, we want to carry on with our iterative process in order to further illuminate the meaning of the concepts involved in this argument.

2.5 Simplifying the Argument

After some further iterations we arrive at a new variant of the original argument: Premises P1 to P4 remain unchanged and a new premise D5 ("x depends for its existence on y := necessarily, x exists only if y exists") is added. D5 corresponds to the 'definition' of ontological dependence as put forth by Lowe in his article (though just for illustrative purposes). As mentioned before, this purported definition was never meant by him to become part of the argument. Nevertheless, we show here how, by assuming the left-to-right direction of this definition, we get in a position to prove the main conclusions without any further assumptions.

axiomatization where

$D5: [\forall^A x y. x \text{ dependsOn } y \rightarrow \Box(E! x \rightarrow E! y)]$

theorem C1: $[\forall^A x. \text{Abstract } x \rightarrow (\exists y. \text{Concrete } y \wedge x \text{ dependsOn } y)]$
using $P3 P4$ by meson

theorem C5: $[\exists^A x. \text{Concrete } x]$
using $P2 P3 P4$ by meson

theorem C10: $[\exists^A x. \text{Godlike } x]$
using *Necessary-def* $P2 P3 P4 D5$ by meson

In this variant we have been able to verify the conclusion of the argument without appealing to the concept of metaphysical explanation. We were able to get by with just the concept of ontological dependence by explicating it in terms of existence and necessity (as suggested by Lowe).

As a side note, we can also prove that the original premise P5 ("No contingent being can explain the existence of a necessary being") directly follows from D5 by redefining metaphysical explanation as the inverse relation of ontological dependence.

abbreviation *explanation::(e \Rightarrow e \Rightarrow wo)* (**infix** *explains*)
where *y explains x* $\equiv x \text{ dependsOn } y$

lemma P5: $[\neg(\exists^A x. \exists^A y. \text{Contingent } y \wedge \text{Necessary } x \wedge y \text{ explains } x)]$
using *Necessary-def* *Contingent-def* $D5$ by meson

In this series of iterations we have reworked the argument so as to get rid of the somewhat obscure concept of metaphysical explanation; we also got some insight into Lowe's concept of ontological dependence vis-a-vis its inferential role in this argument.

There are still some interesting issues to consider. Note that the definitions of existence (*Existence-def*) and being "dependent" (*Dependent-def*) are not needed in any of the highly optimized proofs found by our automated tools. This raises some suspicions concerning the role played by the existence predicate in the definitions of necessariness and contingency, as well as putting into question the need for a definition of being "dependent" linked to the ontological dependence relation. We will see in the following section that our suspicions are justified and that this argument can be dramatically simplified.

2.6 Arriving at a Non-Modal Argument

A new simplified emendation of Lowe's argument is obtained after abandoning the concept of existence and redefining necessariness and contingency

accordingly. As we will see, this variant is actually non-modal and can be easily formalized in first-order predicate logic.

A more literal reading of Lowe's article has suggested a simplified formalization, in which necessariness and contingency are taken as complementary predicates. According to this, our domain of discourse becomes divided in four main categories, as exemplified in the table below:⁸

| | | |
|------------|----------|----------|
| | Abstract | Concrete |
| Necessary | Numbers | God |
| Contingent | Fiction | Stuff |

consts *Necessary::e⇒wo*

abbreviation *Contingent::e⇒wo* **where** *Contingent x ≡ ¬(Necessary x)*

consts *Concrete::e⇒wo*

abbreviation *Abstract::e⇒wo* **where** *Abstract x ≡ ¬(Concrete x)*

abbreviation *Godlike::e⇒w⇒bool* **where** *Godlike x ≡ Necessary x ∧ Concrete x*

consts *dependence::e⇒e⇒wo* (**infix** *dependsOn*)

abbreviation *explanation::(e⇒e⇒wo)* (**infix** *explains*)

where *y explains x ≡ x dependsOn y*

As shown below, we can even define the "dependent" predicate as *primitive*, i.e. bearing no relation to ontological dependence, and still be able to validate the argument. Being "independent" is defined as the negation of being "dependent", as before.

consts *Dependent::e⇒wo*

abbreviation *Independent::e⇒wo* **where** *Independent x ≡ ¬(Dependent x)*

By taking, once again, metaphysical explanation as the inverse relation of ontological dependence and by assuming premises P2 to P5 we can prove conclusion C10.

axiomatization where

P2: [∃ x. Necessary x ∧ Abstract x] and

P3: [∀ x. Abstract x → Dependent x] and

P4: [∀ x. Dependent x → (∃ y. Independent y ∧ x dependsOn y)] and

P5: [¬(∃ x. ∃ y. Contingent y ∧ Necessary x ∧ y explains x)]

theorem C10: [∃ x. Godlike x] using P2 P3 P4 P5 by blast

Note that, in the axioms above, all actualist quantifiers have been changed into non-guarded quantifiers, following the elimination of the concept of

⁸As Lowe explains in the article, "there is no logical restriction on combinations of the properties involved in the concrete/abstract and the necessary/contingent distinctions. In principle, then, we can have contingent concrete beings, contingent abstract beings, necessary concrete beings, and necessary abstract beings."

existence from our argument: Our quantifiers range over *all* beings, because all beings exist. Also note that all modal operators have disappeared; thus, this new variant is directly formalizable in classical first-order logic.

2.7 Modified Modal Argument I

In the following iterations we want to illustrate an approach in which we start our interpretive endeavor with no pre-understanding of the concepts involved. We start by taking all concepts as primitive without providing any definition or presupposing any interrelation between them. We see how we gradually improve our understanding of these concepts in the iterative process of adding and removing axioms and, therefore, by framing their inferential role in the argument.

consts *Concrete*:: $e \Rightarrow wo$
consts *Abstract*:: $e \Rightarrow wo$
consts *Necessary*:: $e \Rightarrow wo$
consts *Contingent*:: $e \Rightarrow wo$
consts *dependence*:: $e \Rightarrow e \Rightarrow wo$ (**infix** *dependsOn*)
consts *explanation*:: $e \Rightarrow e \Rightarrow wo$ (**infix** *explains*)
consts *Dependent*:: $e \Rightarrow wo$
abbreviation *Independent*:: $e \Rightarrow wo$ **where** *Independent* $x \equiv \neg(\textit{Dependent } x)$

In order to honor the original intention of the author, i.e. providing a *modal* variant of St. Anselm's ontological argument, we are required to make a change in Lowe's original formulation. In this variant we have restated the expressions "necessary abstract" and "necessary concrete" as "necessarily abstract" and "necessarily concrete" correspondingly. With this new adverbial reading of the former "necessary" predicate we are no longer talking about the concept of *necessariness*, but of *necessity* instead, so we use the modal box operator (\Box) for its formalization. Note that in this variant we are not concerned with the interpretation of the original ontological argument anymore. We are interested, instead, in showing how our method can go beyond simple interpretation and foster a creative approach to assessing and improving philosophical arguments.

Premise P1 now reads: "God is, by definition, a *necessarily* concrete being."

abbreviation *Godlike*:: $e \Rightarrow wo$ **where** *Godlike* $x \equiv \Box \textit{Concrete } x$

Premise P2 reads: "Some *necessarily* abstract beings exist". The rest of the premises remains unchanged.

axiomatization where

P2: $[\exists x. \Box \textit{Abstract } x]$ **and**

P3: $[\forall x. \textit{Abstract } x \rightarrow \textit{Dependent } x]$ **and**

P4: $[\forall x. \textit{Dependent } x \rightarrow (\exists y. \textit{Independent } y \wedge x \textit{ dependsOn } y)]$ **and**

P5: $[\neg(\exists x. \exists y. \textit{Contingent } y \wedge \textit{Necessary } x \wedge y \textit{ explains } x)]$

Without postulating any additional axioms, C10 ("A *necessarily* concrete being exists") can be falsified by Nitpick.

theorem C10: $[\exists x. \textit{Godlike } x]$
nitpick oops — Countermodel found

An explication of the concepts of necessariness, contingency and explanation is provided below by axiomatizing their interrelation to other concepts. We regard necessariness as being *necessarily abstract* or *necessarily concrete*. We regard explanation as the inverse relation of dependence, as before.

axiomatization where

Necessary-expl: $[\forall x. \textit{Necessary } x \leftrightarrow (\Box \textit{Abstract } x \vee \Box \textit{Concrete } x)]$ **and**
Contingent-expl: $[\forall x. \textit{Contingent } x \leftrightarrow \neg \textit{Necessary } x]$ **and**
Explanation-expl: $[\forall x y. y \textit{ explains } x \leftrightarrow x \textit{ dependsOn } y]$

Without any further constraints, C10 becomes falsified by Nitpick.

theorem C10: $[\exists x. \textit{Godlike } x]$
nitpick oops — Countermodel found

We postulate further modal axioms (using the *Sahlqvist correspondence*) and ask Isabelle's *Sledgehammer* for a proof. Sledgehammer is able to find a proof for C10 which only relies on the modal axiom T ($\Box\varphi \rightarrow \varphi$).

axiomatization where

T-axiom: *reflexive R* **and** — $\Box\varphi \rightarrow \varphi$
B-axiom: *symmetric R* **and** — $\varphi \rightarrow \Box\Diamond\varphi$
IV-axiom: *transitive R* — $\Box\varphi \rightarrow \Box\Box\varphi$

theorem C10: $[\exists x. \textit{Godlike } x]$ **using** *Contingent-expl Explanation-expl*
Necessary-expl P2 P3 P4 P5 T-axiom **by metis**

2.8 Modified Modal Argument II

We start again our interpretive process with no pre-understanding of the concepts involved (by taking them as primitive). We then see how their inferential role gradually becomes apparent in the process of axiomatizing further constraints. We follow on with the adverbial reading of the expression "necessary" as in the previous version.

Another explication of the concepts of necessariness and contingency is provided below. We think that this explication, in comparison to the previous one, better fits our intuitive understanding of necessariness. We now regard necessariness as being *necessarily abstract* or *concrete*, and explanation as the inverse relation of dependence, as before.

axiomatization where

Necessary-expl: $[\forall x. \textit{Necessary } x \leftrightarrow \Box(\textit{Abstract } x \vee \textit{Concrete } x)]$ **and**
Contingent-expl: $[\forall x. \textit{Contingent } x \leftrightarrow \neg \textit{Necessary } x]$ **and**

Explanation-expl: $[\forall x y. y \text{ explains } x \leftrightarrow x \text{ dependsOn } y]$

These constraints are, however, not enough to ensure the argument’s validity as confirmed by Nitpick.

theorem C10: $[\exists x. \text{Godlike } x]$

nitpick oops — Countermodel found

After some iterations, we see that, by giving a more satisfactory explication of the concept of necesariness, we are also required to assume the essentiality of abstractness (as we did in a former iteration) and to restrict the accessibility relation by enforcing its symmetry (i.e. assuming the modal axiom *B*).

axiomatization where

abstractness-essential: $[\forall x. \text{Abstract } x \rightarrow \Box \text{Abstract } x]$ **and**

B-Axiom: *symmetric R*

theorem C10: $[\exists x. \text{Godlike } x]$ **using** *Contingent-expl Explanation-expl*

Necessary-expl P2 P3 P4 P5 abstractness-essential B-Axiom **by metis**

We have chosen to terminate, after this series of iterations, our interpretive endeavor. In each of the previous versions we have illustrated how our understanding of the concepts of necessity/contingency, explanation/dependence and abstractness/concreteness has gradually evolved thanks to the kind of iterative hypothetico-deductive method which has been made possible by the real-time feedback provided by Isabelle’s automated proving tools.

References

- [1] C. Benzmüller, M. Claus, and N. Sultana. Systematic verification of the modal logic cube in Isabelle/HOL. In C. Kaliszyk and A. Paskevich, editors, *PxTP 2015. EPTCS*, volume 186, pages 27–41, Berlin, Germany, 2015.
- [2] C. Benzmüller and L. Paulson. Quantified multimodal logics in simple type theory. *Logica Universalis (Special Issue on Multimodal Logics)*, 7(1):7–20, 2013.
- [3] C. Benzmüller, L. Weber, and B. Woltzenlogel Paleo. Computer-assisted analysis of the Anderson-Hájek controversy. *Logica Universalis*, 11(1):139–151, 2017.
- [4] C. Benzmüller and B. Woltzenlogel Paleo. The inconsistency in Gödels ontological argument: A success story for AI in metaphysics. In *IJCAI 2016*, 2016.

- [5] J. Blanchette, S. Böhme, and L. Paulson. Extending Sledgehammer with SMT solvers. *Journal of Automated Reasoning*, 51(1):109–128, 2013.
- [6] J. Blanchette and T. Nipkow. Nitpick: A counterexample generator for higher-order logic based on a relational model finder. In *Proc. of ITP 2010*, volume 6172 of *LNCS*, pages 131–146. Springer, 2010.
- [7] D. Fuenmayor and C. Benz Müller. Types, Tableaus and Gödel’s God in Isabelle/HOL. *Archive of Formal Proofs*, 2017. This publication is formally verified with Isabelle/HOL.
- [8] E. J. Lowe. Ontological dependence. In E. N. Zalta, editor, *The Stanford Encyclopedia of Philosophy*. Metaphysics Research Lab, Stanford University, spring 2010 edition, 2010.
- [9] J. Moreland. *Debating Christian theism*. Oxford University Press, Oxford New York, 2013.
- [10] T. Nipkow, L. C. Paulson, and M. Wenzel. *Isabelle/HOL — A Proof Assistant for Higher-Order Logic*. Number 2283 in *LNCS*. Springer, 2002.